

## Excel für Microsoft 365

# Tabellenzellen benennen





## Inhaltsverzeichnis

Einleitung.....	2
Benutzerdefinierten Namen festlegen .....	2
Namen aus Auswahl erstellen .....	4
Syntaxregeln für Namen.....	5
Einsatz von benutzerdefinierten Namen.....	6
Die Übersicht behalten.....	10
Konstante Werte und Formeln benennen.....	14
Verschachtelte Namen verwenden .....	16
Relative Bezugsart verwenden.....	17
Benutzerdefinierter Name mit Bezug auf mehrere Blätter.....	18
Benutzerdefinierte Namen für externe Bezüge .....	18
Anzeige von benutzerdefinierten Namen .....	19

## Abbildungsverzeichnis

Abb. 1: <i>Dialogfeld <b>Neuer Name</b></i> .....	2
Abb. 2: <i>Das Namenfeld</i> .....	3
Abb. 3: <i>Die Ausgangstabelle für die finanzmathematische Berechnung</i> .....	4
Abb. 4: <i>Dialogfeld <b>Namen aus Auswahl erstellen</b></i> .....	5
Abb. 5: <i>Das Ergebnis der finanzmathematischen Berechnung</i> .....	5
Abb. 6: <i>Benutzerdefinierten Namen aus einer Liste auswählen</i> .....	7
Abb. 7: <i>Dialogfeld <b>Namen einfügen</b>, beim Dialogfeld <b>Funktionsargumente</b></i> .....	8
Abb. 8: <i>Dialogfeld <b>Namen übernehmen</b></i> .....	9
Abb. 9: <i>Dialogfeld <b>Gehe zu</b></i> .....	10
Abb. 10: <i>Benutzerdefinierten Namen im Namenfeld auswählen</i> .....	11
Abb. 11: <i>Dialogfeld <b>Namen einfügen</b>, beim Symbol <b>In Formel verwenden</b></i> .....	11
Abb. 12: <i>Liste mit allen benutzerdefinierten Namen und zugehörigen Tabellenzellen</i> .....	11
Abb. 13: <i>Dialogfeld <b>Namens-Manager</b></i> .....	12
Abb. 14: <i>Der Einsatz des benutzerdefinierten Namens <b>Ostersonntag</b></i> .....	16
Abb. 15: <i>Der benutzerdefinierte Name wird im Zellbereich angezeigt</i> .....	19

## Einleitung


Wenn Sie Formeln in Ihren Tabellen einsetzen und sich dabei auf den Inhalt von Tabellenzellen beziehen, werden Sie üblicherweise die Zellennamen über die Kombination aus Spaltenbuchstaben und Zeilennummer (also z.B. **F3**) angeben. Dabei können Sie sowohl die relative Schreibweise (z.B. **F3**) oder die absolute Schreibweise (z.B. **\$F\$3**) verwenden (siehe auch Skript **Excel für Microsoft 365 – Relative und absolute Bezüge**). Bei großen und umfangreichen Listen bzw. Tabellen, die neben vielen konstanten Daten auch viele Formeln enthalten, geht bei Verwendung der Originalzellennamen schnell die Übersicht verloren. Zu einem späteren Zeitpunkt können Sie dann die Formeln nur noch schwer nachvollziehen. Sinnvoller ist es, wenn Sie den Tabellenzellen, deren Werte in den Formeln verwendet werden, benutzerdefinierte Namen geben und diese Namen in den Formeln einsetzen. Dadurch behalten Sie viel besser den Überblick über die Daten und die Berechnungen. Insbesondere beim Übertrag von Daten von einem Arbeitsblatt auf ein anderes sind benutzerdefinierte Namen sehr sinnvoll einsetzbar. Sie können sowohl einzelne Tabellenzellen als auch Zellbereiche einen benutzerdefinierten Namen geben. Auch konstante Zahlen (z.B. aus der Physik oder Mathematik) können auf diese Weise definiert und in Formeln eingesetzt werden. Das Skript bezieht sich auf **Excel für Microsoft 365**, gilt aber auch für die älteren Excel-Versionen (auf alle Fälle ab Version 2016).

## Benutzerdefinierten Namen festlegen

Wenn Sie einer Tabellenzelle bzw. einem Zellbereich einen benutzerdefinierten Namen zuweisen wollen, bietet Excel Ihnen mehrere Möglichkeiten an. Zunächst zwei Möglichkeiten:

- Mit Hilfe des Dialogfelds **Neuer Name**
- Mit Hilfe des Namenfelds

Zunächst wird die Vorgehensweise mit dem Dialogfeld gezeigt:

1. Wählen Sie die Tabellenzelle oder markieren den Zellbereich, den Sie benennen wollen.
2. Wählen Sie im Register **Formeln** in der Gruppe **Definierte Namen** das Symbol **Namen definieren** (alternativ: .
3. Im Dialogfeld **Neuer Name** (siehe Abbildung 1) tragen Sie in das Textfeld **Name** den benutzerdefinierten Namen ein. Welche Regeln Sie dabei beachten müssen, lesen Sie im Kapitel **Syntaxregeln für Namen**, Seite 5.





Abb. 1: Dialogfeld **Neuer Name**

4. **Optional**: Wählen Sie in der Liste **Bereich**, ob der benutzerdefinierte Name in der gesamten Arbeitsmappe gültig sein soll oder nur für ein bestimmtes Arbeitsblatt.
5. **Optional**: Tragen Sie in das Textfeld **Kommentar** noch eine zusätzliche Beschreibung für den benutzerdefinierten Namen ein. Hierbei müssen Sie auf keine besonderen Regeln achten. Einzig die Länge ist auf 255 Zeichen beschränkt.
6. **Optional**: Geben Sie im Textfeld **Bezieht sich auf** die Tabellenzelle oder den Zellbereich an, für den der benutzerdefinierte Name gelten soll. Diese Angabe ist aber eigentlich nicht notwendig, wenn Sie bereits in Schritt 1 die Tabellenzelle bzw. den Zellbereich ausgewählt haben, die/den Sie benennen wollen.
7. Bestätigen Sie das Dialogfeld.

Standardmäßig ist ein benutzerdefinierter Name in der gesamten Arbeitsmappe gültig (also für alle Arbeitsblätter; auch wenn manche Arbeitsblätter erst nach Festlegung des benutzerdefinierten Namens eingefügt werden). In diesem Fall darf ein benutzerdefinierter Name nur einmal definiert werden. Wenn Sie also z.B. auf dem Arbeitsblatt **Tabelle1** einer Tabellenzelle den Namen **Anlagebetrag** gegeben haben, dürfen Sie auf keinem anderen Arbeitsblatt einer anderen Tabellenzelle den benutzerdefinierten Namen **Anlagebetrag** zuweisen. Wenn Sie allerdings im Dialogfeld **Neuer Name** (siehe Abbildung 1, Seite 2) in der Liste **Bereich** den Eintrag **Tabelle1** gewählt haben, ist der benutzerdefinierte Name (nehmen wir erneut das Beispiel **Anlagebetrag**) nur für dieses Arbeitsblatt gültig. Sie dürfen also auf einem anderen Arbeitsblatt (z.B. **Tabelle2**) einer Tabellenzelle erneut den benutzerdefinierten Namen **Anlagebetrag** geben (wenn Sie auch hier im Dialogfeld **Neuer Name** in der Liste **Bereich** den Eintrag **Tabelle2** gewählt haben). Diesen Vorgang können Sie für weitere Arbeitsblätter wiederholen. Somit könnte der benutzerdefinierte Name **Anlagebetrag** so oft definiert werden, wie es Arbeitsblätter in der Arbeitsmappe gibt. Allerdings muss von dieser Vorgehensweise abgeraten werden, da Sie mit Sicherheit sehr schnell den Überblick verlieren, wenn ein benutzerdefinierter Name mehr als einmal vorkommt.

Alternativ können Sie den benutzerdefinierten Namen auch über das **Namenfeld** festlegen. Hier die genaue Vorgehensweise:

1. Wählen Sie die Tabellenzelle oder markieren den Zellbereich, den Sie benennen wollen.
2. Klicken Sie in das Namenfeld, das sich links neben der Bearbeitungsleiste befindet (siehe Abbildung 2). **Bitte klicken Sie nicht auf den kleinen grauen Pfeil am rechten Ende des Namenfelds.**





Abb. 2: Das Namenfeld

3. Geben Sie den benutzerdefinierten Namen ein (Regeln siehe Kapitel **Syntaxregeln für Namen**, Seite 5).
4. Bestätigen Sie die Eingabe mit der Eingabetaste ()

**Anmerkung:** Im Gegensatz zum Dialogfeld **Neuer Name** (siehe Abbildung 1, Seite 2) können Sie beim Namenfeld keinen Kommentar angeben und auch nicht bestimmen, ob der Name nur für ein bestimmtes Arbeitsblatt gelten soll.

## Namen aus Auswahl erstellen



Es gibt noch eine alternative Methode, Tabellenzellen benutzerdefinierte Namen zuzuweisen. Dabei müssen Sie die benutzerdefinierten Namen gar nicht selbst eingeben, sondern Excel übernimmt die Namen aus Einträgen in bestimmten Tabellenzellen. Das soll anhand eines Beispiels näher erklärt werden. Nehmen wir an, Sie wollen ein Sparkonto bei einer Bank eröffnen und bei Kontoeröffnung einen festen Geldbetrag (sagen wir als Beispiel **1.000 €**) auf das Konto einzahlen. Außerdem wollen Sie monatlich einen konstanten Betrag (als Beispiel nehmen wir **50 €**) auf das Konto einzahlen. Die Bank gibt Ihnen für dieses Konto einen jährlichen Zinssatz von **1,5%**, allerdings nur, wenn Sie in den nächsten **5 Jahren** kein Geld von dem Konto abheben werden. Sie wollen nun mit Hilfe der finanzmathematischen Funktion **ZW (ZinsWert)** ermitteln, wie viel Geld sich am Ende der 5 Jahre auf Ihrem Konto angesammelt hat. Zunächst werden die Ausgangsdaten in Tabellenzellen auf dem Arbeitsblatt eingetragen. Links neben jeder dieser Tabellenzellen tragen Sie jeweils eine Bezeichnung ein, mit der der entsprechende Zahlenwert genau bezeichnet wird. Die Tabelle<sup>1</sup> könnte folgendes Aussehen haben (siehe Abbildung 3):

	A	B	C
1			
2	<b>Zinssatz:</b>	<b>1,50%</b>	
3	<b>Zeitraum:</b>	<b>5 Jahre</b>	
4	<b>Monatlicher Dauerauftrag:</b>	<b>-50,00 €</b>	
5	<b>Einmalige Einzahlung:</b>	<b>-1.000,00 €</b>	

Abb. 3: Die Ausgangstabelle für die finanzmathematische Berechnung

Die Zahlenwerte für die finanzmathematische Funktion stehen in diesem Beispiel in den Tabellenzellen **B2** bis **B5**. Diese Zellennamen könnten Sie jetzt ohne Probleme in der Formel verwenden. Es ist aber sinnvoller, wenn Sie jeder dieser Tabellenzellen einen benutzerdefinierten Namen zuweisen und diese benutzerdefinierten Namen in der Formel einsetzen. Dadurch wird die Formel transparenter und leichter verständlicher. Jetzt überlegen Sie vielleicht, wie sollen die 4 Tabellenzellen benannt werden und kommen auf die Idee, die Tabellenzelle **B2 Zinssatz** zu nennen, der Tabellenzelle **B3** den Namen **Zeitraum** zu geben, die Tabellenzelle **B4** soll den Namen **Monatlicher Dauerauftrag** bekommen und die Tabellenzelle **B5** soll **Einmalige Einzahlung** heißen. Wenn Sie sich die Tabelle in Abbildung 3 genau anschauen, dann stehen diese vier Namen ja bereits in den Tabellenzellen **A2** bis **A5**. Jetzt überlegen Sie vielleicht: kann ich diese Namen in der Spalte **A** für meine Namensdefinition in der Spalte **B** verwenden? Die Antwort ist: ja. Dazu markieren Sie den Zellbereich mit den Tabellen-

<sup>1</sup> Bei den beiden Geldbeträgen in **B4** und **B5** steht jeweils ein Minuszeichen vor der Zahl. Das bedeutet zunächst, dass Sie (sinnbildlich) Geld an die Bank abgeben. Das Ergebnis der finanzmathematischen Funktion ist dann ein positiver Zahlenwert (sinnbildlich bekommen Sie das Geld von der Bank ausgezahlt).

zellen, wo die Namen eingetragen sind und die Tabellenzellen mit den Zahlenwerten (also die Tabellenzellen, die benannt werden sollen). Das wäre im vorliegenden Beispiel der Zellbereich **A2:B5**. Dann wählen Sie im Register **Formeln** in der Gruppe **Definierte Namen** das Symbol **Aus Auswahl erstellen** (alternativ:  ). Im Dialogfeld **Namen aus Auswahl erstellen** (siehe Abbildung 4) können Sie jetzt bestimmen, wo die Namen im markierten Zellbereich stehen, die als benutzerdefinierte Namen verwendet werden sollen. In diesem Beispiel stehen die Namen *links* von den Tabellenzellen mit den Zahlenwerten. Sie müssen also in diesem Fall nichts verändern (Kontrollkästchen **Linker Spalte** ist bereits aktiviert) und können das Dialogfeld direkt bestätigen. Die Tabellenzellen **B2** bis **B5** bekommen nun die entsprechenden Namen aus den Tabellenzellen **A2** bis **A5** zugewiesen, wobei Excel sich an die Regeln von benutzerdefinierten Zellennamen hält (siehe nachfolgendes Kapitel). Excel passt, falls erforderlich, die Namen selbst an.





Abb. 4: Dialogfeld **Namen aus Auswahl erstellen**

Nun können Sie die benutzerdefinierten Namen in der Formel einsetzen. In diesem Beispiel lautet die Formel (z.B. können Sie die Formel in die Tabellenzelle **B7** eintragen):

**=ZW(Zinssatz/12;Zeitraum\*12;Monatlicher\_Dauerauftrag;Einmalige\_Einzahlung)**

Beachten Sie bitte, dass der Zinssatz durch **12** geteilt und dass der Zeitraum mit dem Wert **12** multipliziert werden muss, da die Funktion **ZW** die Berechnung immer monatsweise durchführt. Das Ergebnis der Berechnung sehen Sie in Abbildung 5.

	A	B	C
1			
2	<b>Zinssatz:</b>	<b>1,50%</b>	
3	<b>Zeitraum:</b>	<b>5 Jahre</b>	
4	<b>Monatlicher Dauerauftrag:</b>	<b>-50,00 €</b>	
5	<b>Einmalige Einzahlung:</b>	<b>-1.000,00 €</b>	
6			
7		<b>4.191,18 €</b>	

Abb. 5: Das Ergebnis der finanzmathematischen Berechnung

## Syntaxregeln für Namen

Wenn Sie einer Tabellenzelle oder einem Zellbereich einen benutzerdefinierten Namen zuweisen wollen, müssen Sie sich an ein paar Syntaxregeln halten, die Excel Ihnen vorgibt:

- Das erste Zeichen des Namens muss ein Buchstabe, ein Unterstrich ( ) oder ein umgekehrter Schrägstrich (**\**) sein.
- Für alle weiteren Zeichen des Namens können Sie Buchstaben, Ziffern, den Unterstrich ( ), den Punkt (.) oder den umgekehrten Schrägstrich (**\**) verwenden.


- Andere Zeichen sind nicht erlaubt. Dazu gehört auch das Leerzeichen (da es in Excel eine ganz spezielle Bedeutung als Schnittmengenoperator hat; siehe Skript **Excel für Microsoft 365 - Relative und absolute Bezüge**, Kapitel **Schnittmengen berechnen**, Seite 8). Verwenden Sie zum Trennen von Wörtern den Unterstrich, den Punkt oder den umgekehrten Schrägstrich.
- Sie können im benutzerdefinierten Namen Groß- oder Kleinbuchstaben verwenden, aber Excel macht keinen Unterschied zwischen Groß- und Kleinschreibung. Wenn Sie eine Tabellenzelle z.B. **ZEITRAUM** genannt haben, dürfen Sie eine andere Tabellenzelle derselben Arbeitsmappe beispielsweise nicht **Zeitraum** oder **zeitraum** nennen.
- Die maximale Länge eines benutzerdefinierten Namens beträgt 255 Zeichen.
- Die Buchstaben **C**, **c**, **R** und **r**<sup>2</sup> dürfen nicht als benutzerdefinierte Namen verwendet werden, da sie alle als Kurzform für das Markieren einer Zeile oder Spalte für die aktuelle markierte Tabellenzelle verwendet werden, wenn Sie diese z.B. im Dialogfeld **Gehe zu**<sup>3</sup> im Textfeld **Verweis** einsetzen.
- Benutzerdefinierte Namen dürfen nicht mit Zellbezügen identisch sein. Z.B. ist der Name **Z3S7** oder **F17** nicht als benutzerdefinierter Name erlaubt.
- Benutzerdefinierte Namen sollten nicht mit den Namen von Excel-Funktionen identisch sein, da es sonst schnell zu Verwechslungen kommen kann<sup>4</sup>.

## Einsatz von benutzerdefinierten Namen

Wenn Sie Tabellenzellen bzw. Zellbereichen benutzerdefinierte Namen zugewiesen haben, können Sie diese benutzerdefinierten Namen nun in beliebig vielen Formeln einsetzen. Allerdings kann es sein, dass ein benutzerdefinierter Name nur für ein Arbeitsblatt gültig ist. Wenn Sie den benutzerdefinierten Namen über das Dialogfeld **Neuer Name** (siehe Abbildung 1, Seite 2) eingegeben und dort in der Liste **Bereich** den Namen eines Arbeitsblatts ausgewählt haben, dann ist der benutzerdefinierte Name nur auf diesem Arbeitsblatt gültig. Verwenden Sie den Namen in einer Formel auf einem anderen Arbeitsblatt, bekommen Sie als Ergebnis den Fehlerwert **#NAME?**. Wollen Sie den benutzerdefinierten Namen trotzdem in einer Formel auf einem anderen Arbeitsblatt einsetzen, müssen Sie dem benutzerdefinierten Namen den Blattnamen zusammen mit dem Ausrufezeichen voranstellen (Beispiel: wenn der benutzerdefinierte Name **Zinssatz** lautet und für das Arbeitsblatt **Tabelle1** festgelegt wurde, können Sie auf einem anderen Arbeitsblatt die Angabe **Tabelle1!Zinssatz** in einer Formel verwenden, wodurch eindeutig geklärt ist, wo der Name **Zinssatz** definiert worden ist).

---

<sup>2</sup> Der Buchstabe **C** bzw. **c** steht für das englische Wort **Column** (dt. Spalte) und **R** bzw. **r** für **Row** (dt. Zeile).

<sup>3</sup> Das Dialogfeld **Gehe zu** bekommen Sie, wenn Sie Register **Start** in der Gruppe **Bearbeiten** das Symbol **Suchen und Auswählen** anklicken und den Befehl **Gehe zu** aufrufen. Alternativ benutzen Sie die Taste .

<sup>4</sup> Grundsätzlich ist es aber erlaubt, einer Tabellenzelle oder einem Zellbereich den Namen einer Excel-Funktion zu geben. Excel kann trotzdem den benutzerdefinierten Namen von dem Namen der Excel-Funktion unterscheiden. Der Name einer Excel-Funktion besitzt immer ein Klammerpaar unmittelbar hinter dem Funktionsnamen, während ein benutzerdefinierter Name kein Klammerpaar hinter dem Namen besitzt. Trotzdem sollten Sie keine Funktionsnamen als benutzerdefinierte Namen verwenden, da es die Übersichtlichkeit deutlich einschränkt.



Normalerweise ist ein benutzerdefinierter Name für die gesamte Arbeitsmappe gültig (**aber wirklich auch nur innerhalb dieser einen Arbeitsmappe**)<sup>5</sup>. Sie können den Namen also in jeder beliebigen Formel einsetzen, egal auf welchem Arbeitsblatt sich die Formel befindet.

Wenn Sie eine Formel in eine Tabellenzelle (oder in einen Zellbereich bei Matrix-Formeln) über die Tastatur eingeben (und nicht über das Dialogfeld **Funktion einfügen**; siehe Skript **Excel für Microsoft 365 – Funktionen (Beispiele)**, Abbildung 4, Seite 29) und in der Formel einen benutzerdefinierten Namen verwenden, müssen Sie während der Eingabe nicht auf die Groß-/Kleinschreibung achten. Sie können den benutzerdefinierten Namen komplett in Kleinbuchstaben eingeben. Sobald Sie die ersten Zeichen eingegeben haben, bekommen Sie eine Auswahlliste, wo Sie neben Excel-Funktionen auch benutzerdefinierte Namen angezeigt bekommen (siehe Abbildung 6). Sie können nun den benutzerdefinierten Namen bequem aus der Liste auswählen (z.B. bewegen Sie das Mausymbol auf den benutzerdefinierten Namen und machen einen Doppelklick). Wenn Sie den benutzerdefinierten Namen nur anklicken, bekommen Sie in einem Infokasten auch den Kommentar angezeigt (wenn Sie in Schritt 5, Seite 3, einen Kommentar eingegeben haben).




Abb. 6: Benutzerdefinierten Namen aus einer Liste auswählen

**Anmerkung:** In der Auswahlliste können Sie eine Excel-Funktion von einem benutzerdefinierten Namen noch durch entsprechende Symbole unterscheiden. Vor einer Excel-Funktion sehen Sie das Symbol und vor dem benutzerdefinierten Namen das Symbol .

Sie können einen benutzerdefinierten Namen bei der Eingabe einer Formel auch aus einer Liste auswählen, anstatt ihn über die Tastatur einzutippen. Wenn Sie in der Formeleingabe an der Stelle sind, wo jetzt ein benutzerdefinierter Name eingegeben werden soll, wählen Sie im Register **Formeln** in der Gruppe **Definierte Namen** das Symbol **In Formel verwenden** und wählen aus der angebotenen Liste den Namen aus. Alternativ benutzen Sie die Taste und wählen den Namen aus dem Dialogfeld **Namen einfügen** (siehe Abbildung 7, Seite 8).



<sup>5</sup> Das stimmt nicht ganz. Soll der benutzerdefinierte Name in einer Formel einer anderen Arbeitsmappe verwendet werden, muss zusätzlich noch ein Verweis auf den Dateinamen der Arbeitsmappe gemacht werden.

Beispiel: `=D:\Test\Mappe1.xlsx!Anlagebetrag`





Abb. 7: Dialogfeld **Namen einfügen**, beim Dialogfeld **Funktionsargumente**

Wenn Sie in einer Formel zum ersten Mal eine neue Funktion einsetzen, werden Sie gegebenenfalls den Funktionsassistenten verwenden (Register **Formeln**, Gruppe **Funktionsbibliothek**, Symbol **Funktion einfügen**). Im Dialogfeld **Funktion einfügen** wählen Sie die Funktion aus und im nachfolgenden Dialogfeld **Funktionsargumente** geben Sie in die Textfelder die entsprechenden Daten für die Funktionsargumente ein. Dabei kann es sich um konstante Daten handeln oder auch um Originalzellennamen. Natürlich können Sie auch benutzerdefinierte Namen angeben. Die Namen können Sie entweder über die Tastatur eintragen, wobei Sie den kompletten Namen eingeben müssen, oder Sie benutzen die Taste **F3** und erhalten das Dialogfeld **Namen einfügen** (siehe Abbildung 7), wo Sie den benutzerdefinierten Namen bequem aus einer Liste auswählen können.



Jetzt kann es in der Praxis auch vorkommen, dass Sie eine Tabelle mit konstanten Daten und Formeln haben, wobei Sie in den Formeln auf Tabellenzellen verweisen, indem Sie die Originalnamen (also z.B. **B4** oder **F12**) verwenden. Zu einem späteren Zeitpunkt entscheiden Sie sich, bestimmten Tabellenzellen benutzerdefinierte Namen zu geben. Natürlich wollen Sie jetzt in den Formeln, wo die Daten dieser Tabellenzellen verwendet werden, die Originalzellennamen durch die benutzerdefinierten Namen ersetzen. Sie könnten natürlich nacheinander die Tabellenzellen mit den Formeln auswählen, in die Zellbearbeitung wechseln (z.B. mit der Taste **F2**) und mühsam jeden Originalzellennamen durch den benutzerdefinierten Namen ersetzen. Das ist für ein paar wenige Formeln vielleicht noch akzeptabel, aber bei sehr vielen Formeln viel zu aufwendig. Das geht einfacher und schneller. Wählen Sie erneut die Tabellenzelle mit der Formel, wo die Originalzellennamen durch die benutzerdefinierten Namen ersetzt werden sollen. Diesmal wechseln Sie aber **nicht** in die Zellbearbeitung. Klicken Sie im Register **Formeln** in der Gruppe **Definierte Namen** neben dem Symbol **Namen definieren** auf den kleinen grauen Pfeil und wählen den Befehl **Namen übernehmen**. Im Dialogfeld **Namen übernehmen** (siehe Abbildung 8, Seite 9) wählen Sie alle benutzerdefinierten Namen aus, die in die Formel übernommen werden sollen. Die beiden Kontrollkästchen **Relative/Absolute Bezugsart ignorieren** und **Zeilen- und Spaltennamen verwenden** sollten aktiviert sein. Sie müssen dabei nicht jeden Originalzellennamen durch einen benutzerdefinierten Namen ersetzen. Wiederholen Sie den ganzen Vorgang für weitere Tabellenzellen mit Formeln, wo Sie die Originalzellennamen durch benutzerdefinierte Namen ersetzen wollen.



Im Dialogfeld **Namen übernehmen** können Sie noch zusätzliche Einstellungen über Kontrollkästchen und der Schaltfläche **Optionen >>** vornehmen. Diese Einstellungen sind aber von untergeordneter Bedeutung und werden daher an dieser Stelle nicht näher beschrieben.




Abb. 8: Dialogfeld **Namen übernehmen**

Sehr praktisch ist der Einsatz von benutzerdefinierten Namen, wenn Sie einen Zelleninhalt (z.B. die Summe der Einnahmen am Monatsende) von einem Arbeitsblatt auf ein anderes übernehmen wollen. Dies soll wieder an einem kleinen Beispiel genauer gezeigt werden. Angenommen, die Summe der Einnahmen befindet sich auf dem Arbeitsblatt mit dem Namen **Januar 2022** und zwar in der Tabellenzelle **B54**. Sie haben dieser Tabellenzelle aber zunächst keinen benutzerdefinierten Namen zugewiesen. Nun soll dieser Ergebniswert in eine andere Tabellenzelle auf einem anderen Arbeitsblatt (aber noch innerhalb derselben Arbeitsmappe) übernommen werden (aber nicht durch Kopieren des Zellinhalts, sondern mit Hilfe eines Verweises über eine Formel). Auf dem anderen Arbeitsblatt (der Name ist an dieser Stelle ohne Bedeutung) wählen Sie die Tabellenzelle aus, wo der Verweis eingetragen werden soll (z.B. die Tabellenzelle **B2**) und geben folgende Formel ein:

**='Januar 2022'!B54**

Zugegeben, die Formel ist nicht besonders lang, aber Sie müssen mehrere Dinge beachten:

- Die Formel enthält den Namen des Arbeitsblatts, wobei der Name in Hochkommas (') eingeschlossen sein muss, wenn er ein oder mehrere Leerzeichen enthält (wie in dem aktuellen Beispiel).
- Der Name der Tabellenzelle auf dem angegebenen Arbeitsblatt.
- Das Ausrufezeichen (!) als Trennzeichen zwischen Blattname und Zellenname.

Wenn Sie nun der Tabellenzelle **B54** auf dem Arbeitsblatt **Januar 2022** einen benutzerdefinierten Namen zugewiesen haben (z.B. **Einnahmen.Januar**), können Sie auf dem anderen Arbeitsblatt in die Tabellenzelle **B2** (aus dem vorherigen Beispiel) folgende Formel schreiben:

**=Einnahmen.Januar**

Zwar ist die Anzahl der einzugebenden Zeichen in beiden Formeln in etwa gleich, aber Sie müssen hinter dem Gleichheitszeichen nur den benutzerdefinierten Namen eingeben (und den können sie auch bequem aus der angebotenen Liste auswählen (siehe Abbildung 6, Seite 7)). Sie brauchen keinen Blattnamen, keine Hochkommas und kein Ausrufezeichen einzugeben.


**Anmerkung:** Wenn Sie einen benutzerdefinierten Namen in einer Formel verwenden, handelt es sich immer um eine absolute Bezugsart (siehe auch Skript **Excel für Microsoft 365 – Relative und absolute Bezüge**).

## Die Übersicht behalten

Wenn Sie innerhalb einer Arbeitsmappe auf verschiedenen Arbeitsblättern eine ganze Reihe an benutzerdefinierten Namen festgelegt haben, verlieren Sie recht schnell die Übersicht. Wenn Sie dann auf einem Arbeitsblatt eine Formel sehen, und bei den Argumenten handelt es sich um benutzerdefinierte Namen, kann es passieren, dass Sie nicht mehr genau wissen, wo sich die dazugehörigen Tabellenzellen bzw. Zellbereiche befinden. Insbesondere wenn Sie von jemand anderem eine Excel-Arbeitsmappe bekommen, wo ebenfalls benutzerdefinierte Namen verwendet werden, ist es eigentlich unmöglich herauszufinden, welche Tabellenzellen bzw. Zellbereiche benannt worden sind und welche nicht. Aber keine Sorge, Excel bietet Ihnen mehrere Möglichkeiten, sich die Tabellenzellen bzw. Zellbereiche zu den benutzerdefinierten Namen anzeigen zu lassen:

- Mit Hilfe des Dialogfelds **Gehe zu**
- Mit Hilfe des Namenfelds
- Liste aller benutzerdefinierter Namen erstellen lassen
- Mit Hilfe des Dialogfelds **Namens-Manager**

Bei allen vier Möglichkeiten spielt es keine Rolle, welches Arbeitsblatt aktuell aktiv ist.

Wenn Sie das Dialogfeld **Gehe zu** (siehe Abbildung 9) verwenden wollen, wählen Sie im Register **Start** in der Gruppe **Bearbeiten** das Symbol **Suchen und Auswählen** und den Befehl **Gehe zu**. (alternativ: ). Im Dialogfeld **Gehe zu** wählen Sie den benutzerdefinierten Namen in der Liste **Gehe zu** und bestätigen das Dialogfeld. Excel wechselt automatisch zum entsprechenden Arbeitsblatt und setzt den Auswahlrahmen auf die zugehörige Tabellenzelle bzw. markiert den zugehörigen Zellbereich.





Abb. 9: Dialogfeld **Gehe zu**

Alternativ können Sie sich die Liste der benutzerdefinierten Namen auch im Namenfeld anzeigen lassen. Klicken Sie im Namenfeld auf den kleinen grauen Pfeil am rechten Ende des Namenfeld (siehe Abbildung 10, Seite 11) und Sie sehen die Liste mit den benutzerdefinierten Namen. Jetzt müssen Sie nur noch den gewünschten Namen anklicken. Wenn die Liste sehr lange benutzerdefinierte Namen enthält, werden diese Namen meistens nicht komplett angezeigt, da die Breite des Namenfelds relativ klein ist. Bewegen Sie das Mausymbol auf die drei kleinen grauen Punkte (⋮) zwischen Namenfeld und Bearbeitungsleiste und vergrößern bzw. verkleinern bei gedrückter linker Maustaste die Breite des Namenfeld (die Breite der Bearbeitungsleiste wird ebenfalls geändert).





Abb. 10: Benutzerdefinierten Namen im Namensfeld auswählen

Bei der dritten Möglichkeit können Sie sich alle benutzerdefinierten Namen und die zugehörigen Tabellenzellen bzw. Zellbereiche auf einem Arbeitsblatt einfügen lassen. Dabei muss es sich nicht zwangsläufig um ein leeres Arbeitsblatt handeln. Es müssen nur zwei Spalten auf dem Arbeitsblatt frei sein. Wählen Sie im ausgesuchten Arbeitsblatt eine freie Tabellenzelle aus, wo Excel mit der Eingabe der benutzerdefinierten Namen beginnen soll. Dann wählen Sie im Register **Formeln** in der Gruppe **Definierte Namen** das Symbol **In Formel verwenden** und den Befehl **Namen einfügen** (alt. ) Im Dialogfeld **Namen einfügen** (siehe Abbildung 11) klicken Sie auf die Schaltfläche **Liste einfügen**. Das Ergebnis sehen Sie in Abbildung 12 (Spaltenbreiten wurden nachträglich angepasst).




Abb. 11: Dialogfeld **Namen einfügen**, beim Symbol **In Formel verwenden**

	A	B	C	D	E	F	G
1	Datum	=41266					
2	Einmalige_Einzahlung	=Tabelle5!\$B\$5					
3	Einnahmen.Quartal	=Tabelle1!\$B\$2:\$B\$5					
4	Fachbereiche	={"Physik"."Mathematik"."Chemie"."Psychologie"."Jura"."Theologie"}					
5	Gesamteinnahmen	=SUMME(Einnahmen)					
6	Monatlicher_Dauerauftrag	=Tabelle5!\$B\$4					
7	Zinssatz	=Tabelle5!\$B\$2					

Abb. 12: Liste mit allen benutzerdefinierten Namen und zugehörigen Tabellenzellen

Bei der vierten Möglichkeit können sie sich alle benutzerdefinierten Namen im Dialogfeld **Namens-Manager** (siehe Abbildung 13, Seite 12) anzeigen lassen. Sie sehen nicht nur die benutzerdefinierten Namen und die zugehörigen Tabellenzellen bzw. Zellbereiche, sondern auch welchen Inhalt diese Tabellenzellen bzw. Zellbereiche enthalten und auch die Kommentare (sofern bei der Erstellung des benutzerdefinierten Namens ein Kommentar in Schritt 5, Seite 3, angegeben wurde). Wählen Sie im Register **Formeln** in der Gruppe **Definierte Namen** das Symbol **Namens-Manager**.





Abb. 13: Dialogfeld **Namens-Manager**

Die Tabellendarstellung im Dialogfeld **Namens-Manager** hat folgenden Aufbau:

Spalte	Bedeutung
<b>Name</b>	In dieser Spalte werden die benutzerdefinierten Namen aufgelistet.
<b>Wert</b>	Hier sehen Sie die aktuellen Werte, die in den Tabellenzellen bzw. Zellbereichen stehen, die zu den benutzerdefinierten Namen gehören. Wenn es sich um einen Zellbereich mit mehreren Werten handelt, werden die Werte in Matrix-Schreibweise mit Punkten bzw. Semikola getrennt und in geschweiften Klammern eingeschlossen dargestellt (siehe auch Skript <b>Excel für Microsoft 365 – Funktionen (Matrix) und dynamische Arrays</b> ).
<b>Bezieht sich auf</b>	In dieser Spalte sehen Sie die zugehörigen Tabellenzellen bzw. Zellbereiche und die Blattnamen, wo sich die Tabellenzellen bzw. Zellbereiche befinden.
<b>Bereich</b>	In dieser Spalte können Sie sehen, ob der benutzerdefinierte Name global für die gesamte Arbeitsmappe gültig ist oder nur für ein bestimmtes Arbeitsblatt.
<b>Kommentar</b>	Wurde in Schritt 5, Seite 3, ein Kommentar für den benutzerdefinierten Namen festgelegt, wird dieser in der letzten Spalte des Namens-Managers angezeigt.

Sie können bei Bedarf die Spaltenbreiten im Namens-Manager anpassen. Bewegen Sie das Mausymbol in der Überschriftzeile (da, wo **Name**, **Wert**, **Bezieht sich auf**, **Bereich** und **Kommentar** steht) auf die Trennlinie zwischen zwei Spalten und verändern mit der Maus die Breite der Spalte bei gedrückter linker Maustaste. Wenn Sie das Mausymbol auf die Trennlinie bewegen und einen Doppelklick ausführen, wird die optimale Spaltenbreite eingestellt. Damit die Spaltenbreitenänderung auch einen

Sinn ergibt, können Sie auch die Breite (und bei Bedarf auch die Höhe) des Dialogfelds über den Rahmen des Dialogfelds mit der Maus anpassen.

Der Namens-Manager bietet noch eine Reihe weiterer Möglichkeiten. Klicken Sie auf die Schaltfläche  und Sie können einen neuen benutzerdefinierten Namen anlegen. Die Vorgehensweise ist dabei identisch mit den Schritten auf Seite 3. Wenn Sie einen benutzerdefinierten Namen ändern wollen (oder die zugehörige Tabellenzelle bzw. der zugehörige Zellbereich), wählen Sie zunächst den benutzerdefinierten Namen in der Auflistung und klicken auf die Schaltfläche  und erhalten das Dialogfeld **Namen bearbeiten** (Aufbau und Nutzung ist identisch mit dem Dialogfeld **Neuer Name** (siehe Abbildung 1, Seite 2)). Sie können natürlich auch jederzeit einen benutzerdefinierten Namen löschen. Wählen Sie den benutzerdefinierten Namen in der Auflistung und klicken auf die Schaltfläche . Excel fragt Sie sicherheitshalber nochmal, ob Sie den benutzerdefinierten Namen auch wirklich löschen wollen. Beim Löschen eines benutzerdefinierten Namens sollten Sie etwas vorsichtig sein. Wird der benutzerdefinierte Name in einer oder mehreren Formeln eingesetzt, erhalten Sie bei diesen Formeln als Ergebnis den Fehlerwert **#NAME?**. Wenn Sie also nicht sicher sind, lassen Sie den benutzerdefinierten Namen lieber im Namens-Manager.


Wenn die Arbeitsmappe sehr viele benutzerdefinierte Namen enthält, kann die Auflistung der Namen im Namens-Manager sehr schnell unübersichtlich werden. Deshalb können Sie die Auflistung filtern. Klicken Sie in der rechten oberen Ecke des Dialogfelds auf die Schaltfläche  und Sie erhalten eine Befehlsliste. Die Befehle haben folgende Bedeutungen:

<i>Befehl</i>	<i>Bedeutung</i>
<b>Filter entfernen</b>	Die Filterung wird entfernt. Es werden wieder alle benutzerdefinierte Namen im Dialogfeld aufgelistet.
<b>Auf Arbeitsblatt ausgelegte Namen</b>	Es werden nur die benutzerdefinierten Namen aufgelistet, die für ein bestimmtes Arbeitsblatt eingeschränkt nutzbar sind.
<b>Auf Arbeitsmappe ausgelegte Namen</b>	Es werden nur die benutzerdefinierten Namen aufgelistet, die in der gesamten Arbeitsmappe benutzt werden können.
<b>Namen mit Fehlern</b>	Es werden nur die benutzerdefinierten Namen aufgelistet, die auf Tabellenzellen bzw. Zellbereiche verweisen, die Fehlerwerte enthalten.
<b>Namen ohne Fehler</b>	Es werden nur die benutzerdefinierten Namen aufgelistet, die auf Tabellenzellen bzw. Zellbereiche verweisen, die keine Fehlerwerte enthalten.
<b>Definierte Namen</b>	Es werden nur die benutzerdefinierten Namen von Tabellenzellen bzw. Zellbereichen aufgelistet.
<b>Tabellennamen</b>	Es werden nur die benutzerdefinierten Namen von Tabellen aufgelistet (siehe Skript <b>Excel für Microsoft 365 – Tabellen</b> ). Nicht zu verwechseln mit Blattnamen.

Sie können die Auflistung im Namens-Manager auch sortieren. Dabei können Sie sich auch aussuchen, nach welcher Spalte (**Name**, **Wert**, **Bezieht sich auf**, usw.) die Sortierung erfolgen soll. Klicken Sie einmal auf den Überschriftenamen der Spalte, die Sie sortiert haben wollen. Dabei werden die Angaben in der betreffenden Spalte auf- oder absteigend sortiert. Klicken Sie nochmals auf die Spaltenüberschrift, wechselt die Sortierreihenfolge.

## Konstante Werte und Formeln benennen

Neben Tabellenzellen und Zellbereichen können Sie auch konstante Werte (Zahl, Text, Datum, Uhrzeit, usw.) mit einem benutzerdefinierten Namen definieren. Beispielsweise können Sie physikalische Konstanten definieren, die Sie dann in physikalischen Berechnungen einsetzen können. Am Beispiel der Konstante für die Lichtgeschwindigkeit soll die Vorgehensweise beim Definieren von Konstanten näher gezeigt werden.

1. Benutzen Sie z.B. die Tastenkombination  um das Dialogfeld **Neuer Name** (siehe Abbildung 1, Seite 2) aufzurufen.
2. Tragen Sie in das Textfeld **Name** den Namen für die Konstante ein (in diesem Beispiel wird der Name **Lichtgeschwindigkeit** genommen<sup>6</sup>).
3. Löschen Sie den Inhalt im Textfeld **Bezieht sich auf** und geben den konstanten Wert (in diesem Beispiel **299792,458**) ein und bestätigen das Dialogfeld.

Jetzt können Sie den Namen **Lichtgeschwindigkeit** wie jeden anderen benutzerdefinierten Namen in beliebigen Formeln einsetzen.

**Anmerkung:** Benutzerdefinierte Namen gelten nur innerhalb der Arbeitsmappe. Das gilt auch bei Konstanten. Wenn Sie bestimmte Konstanten global in sehr vielen Arbeitsmappen einsetzen wollen, müssen Sie dafür benutzerdefinierte Funktionen mit der Programmiersprache **VBA** (Visual Basic for Applications; siehe Skript **Excel für Microsoft 365 – Benutzerdefinierte Funktionen**) erstellen.

Sie können auch komplette Listen als konstante Daten bereitstellen. Die Listeneinträge können Sie später z.B. als Spaltenüberschriften für Tabellen verwenden. Als Beispiel soll eine Liste mit Adressfeldern für Personendaten (also *Name, Vorname, Straße, PLZ, Ort*, usw.; wie viele Einträge es genau gibt, können Sie selbst festlegen) angelegt werden. Analog zum Beispiel mit der Lichtgeschwindigkeit öffnen Sie das Dialogfeld **Neuer Name**, geben in das Textfeld **Name** einen benutzerdefinierten Namen ein (z.B. **Adresszeile**) und geben in das Textfeld **Bezieht sich auf** folgenden Eintrag ein:

```
={"Name"."Vorname"."Straße"."PLZ"."Ort".Tel-Nr"."E-Mail-Adresse"}
```



Wie Sie an dem Beispiel sehen können, werden die Listenelemente in ein geschweiftes Klammerpaar ( { } ) eingeschlossen. Jedes Listenelement ist zudem in doppelte Anführungszeichen ( " ) eingeschlossen. Die Listenelemente sind durch Punkte ( . ) getrennt. Weitere Informationen hierzu finden Sie im Skript **Excel für Microsoft 365 - Matrix-Funktionen**.

Wenn Sie jetzt diese Listenelemente in einem (neuen) Arbeitsblatt brauchen, wählen Sie eine Tabellenzelle, ab der die Daten eingefügt werden sollen. Dann geben Sie die Formel

```
=Adresszeile
```


<sup>6</sup> Wenn Sie jetzt daran denken, dass es für die meisten physikalischen Konstanten eine Abkürzung gibt (die Lichtgeschwindigkeit wird mit dem Buchstaben *c* gekennzeichnet) und Sie wollen diese Abkürzung als benutzerdefinierten Namen verwenden, dann muss daran erinnert werden, dass die Buchstaben **C**, **c**, **R** und **r** nicht für benutzerdefinierte Namen genommen werden dürfen.



ein und bestätigen die Eingabe mit der Eingabetaste ()<sup>7</sup>. Obwohl es sich eigentlich um eine Matrixberechnung handelt, müssen Sie nicht die Tastenkombination  verwenden. In diesem Fall interpretiert Excel für Microsoft 365 die Eingabe als dynamisches Array (siehe Skript **Excel für Microsoft 365 – Funktionen (Matrix) und dynamische Arrays**, Kapitel **Dynamische Arrays**, Seite 12).

Es kann eventuell vorkommen, dass Sie die Elemente dieser Liste auch mal in Tabellenzellen in einer Spalte (also untereinander) eingeben möchten. In diesem Fall wählen Sie ebenfalls eine Tabellenzelle, ab der die Einträge nach unten durchgeführt werden sollen und geben folgende Formel ein:

**=MTRANS(Adresszeile)**

Die Eingabe schließen Sie ebenfalls nur mit der Eingabetaste () ab (Stichwort: *dynamisches Array*). Mit Hilfe der Funktion **MTRANS** werden die Listenelemente aus der Zeilenanordnung in die Spaltenanordnung transponiert.

**Anmerkung:** Wenn Sie konstante Listen erstellen wollen, wobei die Listenelemente später immer in einer Spalte untereinanderstehen sollen, müssen Sie im Textfeld **Bezieht sich auf** (Dialogfeld **Neuer Name**) zwischen den Listeneinträgen anstelle der Punkte jeweils das Semikolon (;) verwenden.

Sie können im Dialogfeld **Neuer Name** im Textfeld **Bezieht sich auf** hinter dem Gleichheitszeichen eine Formel eingeben. Die Formel kann auch Zellennamen (z.B. **B3**) bzw. benutzerdefinierte Zellennamen enthalten. Wird der Name dann in einer Formel verwendet (die sich in einer Tabellenzelle befindet), wird zunächst die zum Namen gehörende Formel ausgewertet und das Ergebnis dann in der Formel weiterverwendet, die in der Tabellenzelle steht. Dazu erneut ein kleines Beispiel:

Excel enthält zwar eine ganze Reihe an Datumsfunktionen, aber mit keiner ist es möglich, das Datum für den Ostersonntag zu berechnen. Dafür müssen Sie eine verhältnismäßig komplexe Formel erstellen (unter Zuhilfenahme von verschiedenen Excel-Funktionen). Die Formel lautet<sup>7</sup> (in diesem Beispiel wird vorausgesetzt, dass sich die Jahreszahl in der Tabellenzelle **B1** befindet):

**=KÜRZEN(DATUM(\$B\$1;3;56-REST(REST(\$B\$1;19)\*10,63+5;29))/7)\*7+1**

Diese Funktion jedes Mal einzugeben, wenn sie in der Arbeitsmappe gebraucht wird, ist zu aufwendig, zumal hierbei auch sehr leicht Eingabefehler passieren können. Deshalb wird die Formel im Dialogfeld **Neuer Name** eingeben und als Name bietet sich **Ostersonntag** an. Nun kann der Name überall auf dem Arbeitsblatt eingesetzt werden. Der Name kann auch in anderen Formeln eingesetzt werden (z.B. zum Berechnen der Feiertage *Karfreitag*, *Ostermontag*, *Christi Himmelfahrt*, *Pfingsten* und *Fronleichnam*; diese Feiertage ergeben sich durch Subtraktion bzw. Addition von genau vorgegebenen Anzahl von Tagen). Abbildung 14, Seite 16, zeigt die Feiertage für das Jahr 2022.

<sup>7</sup> Die Formel funktioniert nur, wenn im Dialogfeld **Excel-Optionen** in der Kategorie **Erweitert** in der Gruppe **Beim Berechnen dieser Arbeitsmappe** das Kontrollkästchen **1904-Datumswerte verwenden** **deaktiviert** ist.

	A	B	C	D	E	F	G
1	Jahr:	2022		Karfreitag:	Freitag, 15.04.2022		Formel in E1: =Ostersonntag-2
2				Ostersonntag:	Sonntag, 17.04.2022		Formel in E1: =Ostersonntag
3				Ostermontag:	Montag, 18.04.2022		Formel in E1: =Ostersonntag+1
4				Christi Himmelfahrt:	Donnerstag, 26.05.2022		Formel in E1: =Ostersonntag+39
5				Pfingstsonntag:	Sonntag, 05.06.2022		Formel in E1: =Ostersonntag+49
6				Pfingstmontag:	Montag, 06.06.2022		Formel in E1: =Ostersonntag+50
7				Fronleichnam:	Donnerstag, 16.06.2022		Formel in E1: =Ostersonntag+60

Abb. 14: Der Einsatz des benutzerdefinierten Namens **Ostersonntag**

**Anmerkung:** Die Formel für die Berechnung des Ostersonntags liefert kein wirkliches Datum, sondern nur eine Zahl. Diese muss dann noch mit einem Zahlenformat als Datum dargestellt werden. In Abbildung 14 wurde das Zahlenformat **TTTT, T. MMMM JJJJ** für den Zellbereich **E1:E7** verwendet.

Sie können jetzt in der Tabellenzelle **B1** die Jahreszahl ändern (bis maximal 2099) und sehen dann im Zellbereich **E1:E7** die Datumsangaben für die entsprechenden Feiertage. Bedenken Sie aber bitte immer, dass der benutzerdefinierte Name nur in der aktuellen Arbeitsmappe gültig ist. Wenn Sie die Formel für die Berechnung des Osterdatums global in beliebigen Arbeitsmappen einsetzen wollen, müssen Sie eine benutzerdefinierte Funktion erstellen (siehe Skript **Excel für Microsoft 365 – Funktionen (benutzerdefiniert)**, Kapitel **Berechnung des Ostersonntags**, Seite 5).

## Verschachtelte Namen verwenden

Excel bietet Ihnen die Möglichkeit, einen benutzerdefinierten Namen in Bezug auf einen anderen benutzerdefinierten Namen zu verwenden. Das klingt etwas verwirrend, daher hier wieder ein konkretes Beispiel. Angenommen Sie haben in einem beliebigen Arbeitsblatt im Zellbereich **B2:B13** die Einnahmen für die Monate *Januar* bis *Dezember* eines Jahres eingetragen (in der Tabellenzelle **B1** steht *Einnahmen* als Spaltenüberschrift). Nun soll die Summe der Einnahmen ermittelt werden. Das Ergebnis der Berechnung brauchen Sie aber jetzt mehrmals und evtl. auch in verschiedenen Arbeitsblättern der Arbeitsmappe. Zunächst markieren Sie den Zellbereich **B2:B13** und geben ihm einen benutzerdefinierten Namen (z.B. **Einnahmen**). Jetzt könnten Sie in jeder beliebigen Tabellenzelle (unabhängig vom Arbeitsblatt) die Formel

**=SUMME(Einnahmen)**

eintragen. Anstelle der Formel können Sie aber auch einen weiteren benutzerdefinierten Namen verwenden. Öffnen Sie dazu das Dialogfeld **Neuer Name**, tragen in das Textfeld **Name** einen benutzerdefinierten Namen ein (z.B. **Gesamteinnahmen**) und geben in das Textfeld **Bezieht sich auf** die Formel **=SUMME(Einnahmen)** ein und bestätigen das Dialogfeld. Wenn Sie jetzt die Summe der Einnahmen in verschiedenen Tabellenzellen benötigen, geben Sie in diese Tabellenzellen die Formel

**=Gesamteinnahmen**

ein und erhalten die *Summe der Einnahmen*. Zugegeben, die Ersparnis bei der Formeleingabe ist in diesem Beispiel nicht besonders hoch. Wenn es sich aber um eine komplexere Berechnung (z.B. mit mehreren ineinander verschachtelten Funktionen) handelt, müssten Sie die Formel also nur einmal

eingeben (im Textfeld **Bezieht sich auf**) und nicht mehrere Male in verschiedenen Tabellenzellen der Arbeitsmappe.

## Relative Bezugsart verwenden

In der Anmerkung auf Seite 9 können Sie nachlesen, dass, wenn Sie einer Tabellenzelle einen benutzerdefinierten Namen zuweisen, es sich dabei immer um einen absoluten Bezug handelt. Wenn Sie diesen benutzerdefinierten Namen in irgendwelchen Formeln verwenden, ist immer der Bezug auf diese eine ganz bestimmte Tabellenzelle festgelegt. Excel bietet aber auch die Möglichkeit an, relative Bezüge (oder auch Mischformen) bei der Verwendung von benutzerdefinierten Namen einzusetzen. Auch in diesem Fall wird für eine bessere Verständlichkeit ein konkretes Beispiel genommen.

In einer Tabelle sind in Spalte **A** (ab Tabellenzelle **A2**) die Einnahmen und in Spalte **B** (ab Tabellenzelle **B2**) die Ausgaben eingetragen. In Spalte **C** (genauer: ab Tabellenzelle **C2**) soll nun jeweils der Differenzbetrag zwischen Einnahmen und Ausgaben (genauer **=B2-A2**, **=B3-A3**, **=B4-A4**, usw.) ermittelt werden. Dabei soll ein benutzerdefinierter Name eingesetzt werden. Öffnen Sie das Dialogfeld **Neuer Name**, geben in das Textfeld **Name** einen beliebigen benutzerdefinierten Namen ein (z.B. **Differenzbetrag**) und in das Textfeld **Bezieht sich auf** schreiben Sie die Formel **= $\$$ B2- $\$$ A2** (beachten Sie bitte, dass nur vor den Spaltenbuchstaben ein Dollarzeichen angegeben ist; die Zeilennummer ist also relativ) und bestätigen das Dialogfeld. Nun wählen Sie die Tabellenzelle **C3** aus und geben als Formel ein:

**=Differenzbetrag**

Sie können anschließend den Eintrag von Tabellenzelle **C2** mit Hilfe des *automatischen Ausfüllens* (siehe Skript **Excel für Microsoft 365 – Automatisches Ausfüllen**) in die Tabellenzellen **C3**, **C4**, **C5**, usw. kopieren.

Bei einem zweiten Beispiel stehen in einem Zellbereich (in diesem Beispiel ist das der Zellbereich **A1:J5**) Zahlen, wobei immer die Werte in der jeweiligen Spalte aufsummiert werden sollen. Die Ergebnisse werden in der Zeile **7** eingetragen (Zellbereich **A7** bis **J7**). Setzen Sie den Auswahlrahmen auf die Tabellenzelle **A7** und öffnen erneut das Dialogfeld **Neuer Name**. In das Textfeld **Name** tragen Sie einen Namen Ihrer Wahl ein (in diesem Beispiel wird der Name **Bereich** genommen). In das Textfeld **Bezieht sich auf** tragen Sie ein

**=A1:A5**

und bestätigen dann das Dialogfeld. Tragen Sie jetzt in die (immer noch ausgewählte) Tabellenzelle **A7** die Formel

**=SUMME(Bereich)**

ein und bestätigen die Eingabe. Die Summen der anderen Spalten (**B** bis **J**) können Sie durch Kopieren der Formel aus der Tabellenzellen **A7** mit Hilfe des Verfahrens *automatisches Ausfüllen* in die Tabellenzellen **B7** bis **J7** übertragen.

## Benutzerdefinierter Name mit Bezug auf mehrere Blätter

Excel erlaubt es Ihnen auch, benutzerdefinierte Namen mit Bezug auf mehrere Blätter festzulegen. Auch hierfür direkt ein konkretes Beispiel. Angenommen Sie haben verschiedene Messreihen durchgeführt und die Ergebnisse in verschiedenen Arbeitsblättern eingetragen. Dabei haben Sie aber auf jedem Arbeitsblatt immer denselben Zellbereich verwendet (**das ist wichtig!**). Beispielsweise stehen die Messwerte auf allen Arbeitsblättern im Zellbereich **A1:B10**. Als Beispiel wird angenommen, es gibt drei Messreihen, also drei Arbeitsblätter. Diese haben die Blattnamen *Messreihe1*, *Messreihe2* und *Messreihe3* (alle Blattnamen ohne Leerzeichen). Sie möchten gerne den Mittelwert aus allen Messdaten aller Messreihen ermitteln. Zunächst wird der benutzerdefinierte Name festgelegt. Im Dialogfeld **Neuer Name** tragen Sie in das Textfeld **Name** einen beliebigen benutzerdefinierten Namen ein (z.B. **Messreihen**) und in das Textfeld **Bezieht sich auf** **=Messreihe1:Messreihe3!\$A\$1:\$B\$10** und bestätigen das Dialogfeld. Auf einem weiteren Arbeitsblatt (Blattname ist ohne Bedeutung) wählen Sie die Tabellenzelle aus, wo der Mittelwert der Messdaten eingetragen werden soll, und geben folgende Formel ein:

```
=MITTELWERT(Messreihen)
```

## Benutzerdefinierte Namen für externe Bezüge

Sie können auch Bezüge auf Tabellenzellen oder Zellbereiche festlegen, die sich in anderen Arbeitsmappen befinden. Damit der Bezug auch eindeutig ist, muss neben dem Zellennamen auch der Blattname und der Dateiname der Arbeitsmappe angegeben werden. Dabei müssen Sie noch unterscheiden, ob die Arbeitsmappe, für die der Bezug gelten soll, in Excel geöffnet ist oder nicht. Nehmen wir einmal an, Sie wollen in der aktuellen Arbeitsmappe einen Bezug für die Tabellenzelle **C5** (besitzt den benutzerdefinierten Namen **Gesamtergebnis**) herstellen, die sich auf dem Arbeitsblatt **Jahresabschluss** in der Arbeitsmappe **Geschäftsjahr 2022.xlsx** befindet. Die Arbeitsmappe ist auf dem Festplattenlaufwerk **D:** im Ordner **Geschäftsdaten** abgelegt. In der aktuellen Arbeitsmappe wählen Sie nun eine Tabellenzelle auf einem beliebigen Arbeitsblatt aus und geben folgende Formel ein (die Arbeitsmappe **Geschäftsjahr 2022.xlsx** ist aktuell geöffnet):

```
=[Geschäftsjahr 2022.xlsx]Jahresabschluss!C5
```

Wenn die Arbeitsmappe **nicht** geöffnet ist, sieht die Formel folgendermaßen aus:

```
='D:\Geschäftsdaten\[Geschäftsjahr 2022.xlsx]Jahresabschluss'!C5
```

Es wird ja in diesem Beispiel davon ausgegangen, dass die Tabellenzelle **C5** den benutzerdefinierten Namen **Gesamtergebnis** besitzt. In diesem Fall lautet die Formel (die Arbeitsmappe **Geschäftsjahr 2022.xlsx** ist geöffnet):

```
='Geschäftsjahr 2022.xlsx'!Gesamtergebnis
```

Ist die Arbeitsmappe **Geschäftsjahr 2022.xlsx** **nicht** geöffnet, lautet die Formel:

```
='D:\Geschäftsdaten\Geschäftsjahr 2022.xlsx'!Gesamtergebnis
```

## Anzeige von benutzerdefinierten Namen

Zum Abschluss dieses Themas noch eine kleine Besonderheit bei der Anzeige von benutzerdefinierten Namen. Diese Besonderheit gilt aber nur für Zellbereiche, denen Sie benutzerdefinierte Namen zugewiesen haben. Wenn Sie die Zoomgröße auf **39%** oder kleiner einstellen, wird in dem Zellbereich, dem Sie den benutzerdefinierten Namen zugewiesen haben, der Name sichtbar angezeigt. Wird die Zoomgröße auf einen Wert größer als **39%** eingestellt, ist der benutzerdefinierte Name nicht mehr zu sehen. In Abbildung 15 wurde einem umfangreichen Zellbereich ein benutzerdefinierter Name (in diesem Beispiel der Name **Zellbereich**) zugewiesen und anschließend die Zoomgröße auf **30%** eingestellt. Da bei einer so kleinen Zoomgröße der Zellinhalt eh nicht mehr lesbar ist, wurde auf einen Inhalt der Tabellenzellen verzichtet. Um den Zellbereich besser erkennen zu können, wurde hierfür die Füllfarbe Gelb genommen.



Abb. 15: Der benutzerdefinierte Name wird im Zellbereich angezeigt